# L'intégrale définie de Riemann:

$$\int_a^b f(x)\,\mathrm{d}x = \lim_{\max \Delta x_i \to 0} \left( \sum_{i=1}^{n} f(c_i)\,\Delta x_i \right)$$

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2002. L'objectif principal de de document Maple est de montrer que la définition de la somme intégrale au sens de Riemann n'exige pas que la partition de l'intervalle soit régulière ni même que le choix du représentant dans chaque sous-intervalle créé par la partition soit systématiquement la borne inférieure ou le milieu ni même systématiquement la borne supérieure. Après la présentation de la macro-commande *RiemannSum* de la sous-bibliothèque *Sudent[Calculus1]* dont les arguments *method* et *output* permettent de développer et d'illustrer des sommes intégrales dans le cas de partitions régulières (ou non régulières), il vous sera présenté la macro-commande ***plotriemann*** de la bibliothèque ***riemann*** que j'ai écrite pour mes élèves. Cette macro-commande retourne la valeur numérique et la représentation graphique de sommes intégrales avec des partitions pouvant être non-régulières et avec un choix de représentants pouvant être quelconques et même non choisis systématiquement allant même à être choisis aléatoirement dans des partitions qui sont elles-mêmes aléatoires . La macro-commande ***plotriemann*** retrourne aussi, pour des partitions quelconques, les valeurs des sommes inférieure et supérieure de Riemann ainsi que leur représentation graphique. (les mêmes options qu'on retrouve maintenant avec *RiemannSum*).

Ce document se termine par l'application de la méthode de *Monte Carlo* pour estimer la valeur d'une intégrale définie.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

## Initialisation

```
> restart;
```

```
> with(plots,display,setoptions);
  setoptions(labels=[x,y],tickmarks=[12,12],
             axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8],
             legendstyle=[font=["TIMES", 8]],
             size=[400,400]);
```

$$[display, setoptions] \tag{1}$$

## La notation $\Sigma$ (sigma majuscule)

Le symbole sigma majuscule $\sum$ est utilisée afin de symboliser une addition de plusieurs termes.

```
> `1+2+3+4+5+6+7+8+9+10`=Sum(i,i=1..10);
  ``=sum(i,i=1..10);
```

$$1+2+3+4+5+6+7+8+9+10=\sum_{i=1}^{10} i$$

$$=55 \quad \textbf{(2)}$$

La lettre $i$ est l'indice de la sommation. La notation $\sum_{i=1}^{10} i$ s'énonce comme suit: somme, pour $i$ variant de 1 jusqu'à 10, de (la portée) $i$.

- Le choix de la lettre spécifiant l'indice de sommation est arbitraire et ne sert qu'à permettre le développement de la portée du symbole de sommation. Les lettres $i, j, k$ et $l$ sont les lettres qui sont habituellement choisis.
- Par convention, la portée du symbole de sommation est limité par un signe d'addition ou un signe de soustraction. S'il faut inclure une addition et/ou une soustraction de termes dans la portée, il faut utiliser des parenthèses rondes afin d'inclure ces opérations dans la portée.

Ainsi, d'une part, la notation $\sum_{k=1}^{10} k$ représente tout autant la même somme 1+2+3+4+5+6+7+8+9+10 que $\sum_{i=1}^{10} i$.

```
> Sum(k,k = 1 .. 10) = Sum(i,i = 1 .. 10);
  value(%);
```

$$\sum_{k=1}^{10} k=\sum_{i=1}^{10} i$$

$$55=55 \quad \textbf{(3)}$$

D'autre part, $\sum_{k=1}^{10} k+10 \neq \sum_{k=1}^{10} (k+10)$.

```
> Sum(k,k = 1 .. 10)+10 <> Sum(k+10,k = 1 .. 10);
  value(%);
```

$$\left(\sum_{k=1}^{10} k\right)+10 \neq \sum_{k=1}^{10} (k+10)$$

$$65 \neq 155 \quad \textbf{(4)}$$

Finalement, il est possible de *ré-indexer* l'indice de sommation sans changer la valeur de la somme:

- pour augmenter de $r$ unités les valeurs initiale et finale de l'indice de sommation, il suffit de soustraire $r$ unités à tous les indices de sommation dans la portée
- pour diminuer de $r$ unités les valeurs initiale et finale de l'indice de sommation, il suffit d'additionner $r$ unités à tous les indices de sommation dans la portée

Par exemples,

```
> Sum(k,k = 1 .. 10)=Sum(k+8,k = -7 .. 2);
  value(%);
```

$$\sum_{k=1}^{10} k = \sum_{k=-7}^{2} (k+8)$$

$$55 = 55 \qquad \textbf{(5)}$$

```
> Sum(k,k = 1 .. 10)=Sum(k-15,k = 16 .. 25);
  value(%);
```

$$\sum_{k=1}^{10} k = \sum_{k=16}^{25} (k-15)$$

$$55 = 55 \qquad \textbf{(6)}$$

# Somme intégrale ou somme de Riemann

Soit $f$ une fonction définie sur un intervalle $[a,b]$.
Soit $P = \{x_0, x_1, x_2, x_3, \ldots, x_{n-1}, x_n\}$ une partition de l'intervalle $[a,b]$ où

$$x_0 = a < x_1 < x_2 < x_3 < \ldots < x_{i-1} < x_i < x_{i+1} < \ldots < x_{n-1} < x_n = b.$$

La partition $P$ détermine donc $n$ sous-intervalles $(i = 1..n)$ $[x_{i-1}, x_i]$ telle que $x_0 = a$ et $x_n = b$.

Soit $c_i \in [x_{i-1}, x_i]$ un représentant arbitraire du $i^{ième}$ sous-intervalle de l'intervalle $[a,b]$.

Soit $\Delta x_i = |x_i - x_{i-1}|$, la largeur du $i^{ième}$ sous-intervalle $[x_{i-1}, x_i]$. $(\Delta x_i > 0)$

On définit, sur l'intervalle $[a,b]$, une somme intégrale (somme de Riemann) comme étant la somme

$$SR_{[a,b]} = \sum_{i=1}^{n} f(c_i)\,\Delta x_i. \quad \text{où} \quad c_i \in [x_{i-1}, x_i]$$

D'une part, la définition de $SR_{[a,b]}$ n'implique pas que la partition $P$ soit régulière, c'est-à-dire que pour tout indice $i$, nous n'avons pas nécessairement $\Delta x_i = \dfrac{b-a}{n}$. D'autre part, on peut créer, avec une même partition $P$, autant de sommes intégrales différentes qu'il y a de manières de choisir le représentant $c_i$ dans chaque sous-intervalle $[x_{i-1}, x_i]$.

Dans le cas bien particulier où la partition $P$ est régulière et lorsque le représentant $c_i$ de chaque sous-intervalle $[x_{i-1}, x_i]$ est choisi systématiquement comme étant la borne inférieure, c'est-à-dire lorsque $c_i = x_{i-1}$, il est assez facile, dans le cas de certaines fonctions, de simplifier la somme intégrale $SR_{[a,b]}$. Dans ce cas, nous noterons cette somme intégrale $SR_{gauche_{[a,b]}}$. La macro-commande *ReimannSum* dont l'argument *method=left* va nous permettre de développer une somme $SR_{gauche_{[a,b]}}$.

```
> with(Student[Calculus1],RiemannSum);
```

$$[RiemannSum] \qquad \textbf{(7)}$$

Considérons la fonction f définie par $f(x) = 1 - x^2$ et soit l'intervalle [0,3]. La fonction f est bien définie sur l'intervalle [0,3]. En considérant une partition $P$ régulière de l'intervalle [0, 3] créons six sous-intervalles de même largeur $\Delta x_i = \frac{3-0}{6} = \frac{1}{2}$ , obtenons la somme intégrale en prenant systématiquement comme représentant $c_i$, la borne inférieure de chaque sous-intervalles $\left[x_{i-1}, x_i\right]$.

Saisissons d'abord la fonction f.

```
> f:=x->1-x^2;
```

$$f := x \mapsto 1 - x^2 \tag{8}$$

Obtenons la somme intégrale $SR_{gauche_{[0,\,3]}}$ à l'aide de la macro-commande *ReimannSum*.

```
> SR[gauche][[0,3]]=RiemannSum(f(x),x=0..3,method=left,partition = 6,
  output=sum);
```

$$\left(SR_{gauche}\right)_{[0,\,3]} = \frac{\left(\sum_{i=0}^{5}\left(-\frac{i^2}{4}+1\right)\right)}{2} \tag{9}$$

```
> value((9));
```

$$\left(SR_{gauche}\right)_{[0,\,3]} = -\frac{31}{8} \tag{10}$$

La somme intégrale $SR_{gauche_{[0,\,3]}}$ peut être représentée géométriquement en interprétant chaque terme de la somme intégrale $\sum_{i=0}^{5} f\left(c_i\right)\Delta x_i$ comme le produit des deux dimensions d'un rectangle de largeur (base) $\Delta x_i$ et de longueur (hauteur) $\mathrm{f}\left(c_i\right)$ . On donnera ici un sens «algébrique» aux termes dimensions. Bien que la dimension largeur $\Delta x_i$ soit toujours positive, la dimension longueur $\mathrm{f}\left(c_i\right)$ , quant à elle, possède un signe algébrique, soit celle de la fonction f. De sorte que, lorsque $\mathrm{f}\left(c_i\right) > 0$, le rectangle représentatif du terme $\mathrm{f}\left(c_i\right)\Delta x_i$ sera illustré au-dessus de l'axe des $x$ (partie positive de l'axe de $y$) et lorsque $\mathrm{f}\left(c_i\right) < 0$, le rectangle représentatif le sera au-dessous de l'axe des $x$ (partie négative de l'axe de $y$).

La macro-commande *RiemannSum* de la bibliothèque *Student*[*Calculus1*] permet de représenter graphiquement cette somme intégrale.

```
> RiemannSum(f(x),x=0..3,
    method=left,
    partition = 6,
    output=plot,
    boxoptions = [filled = [color = turquoise, transparency = .5]]);
```




Une approximation de $\int_0^3 f(x)\,\mathrm{d}x$ par une somme gauche de Rie

Obtenons ensuite les sommes intégrales lorsque le représentant $c_i$ est systématiquement choisi comme étant le milieu de chaque sous-intervalle $\left[x_{i-1}, x_i\right]$ et lorsque qu'il est systématiquement choisi comme étant la borne supérieure de chaque sous-intervalle à l'aide des *method*: *midpoint* et *right*.

Développons et représentons graphiquement les sommes intégrales $SR_{milieu_{[0,\,3]}}$ et $SR_{droite_{[0,\,3]}}$.

```
> SR[milieu][[0,3]]=RiemannSum(f(x),x=0..3,method=midpoint,partition =
  6,output=sum);;
  value(%);
```

$$\left(SR_{milieu}\right)_{[0,\,3]} = \frac{\left(\sum_{i=0}^{5}\left(-\left(\frac{i}{2}+\frac{1}{4}\right)^{2}+1\right)\right)}{2}$$

$$\left(SR_{milieu}\right)_{[0,\,3]} = -\frac{95}{16} \qquad \textbf{(11)}$$

```
> RiemannSum(f(x),x=0..3,
    method=midpoint,
    partition = 6,
    output=plot,
    boxoptions = [filled = [color = turquoise, transparency = .5]]);
```




Une approximation de $\int_0^3 f(x)\,\mathrm{d}x$ par une somme médiane de R

Finalement, obtenons $SR_{droite_{[0,\,3]}}$.

```
> SR[droite][[0,3]]=RiemannSum(f(x),x=0..3,method=right,partition = 6,
  output=sum);
  value(%);
```

$$\left(SR_{droite}\right)_{[0,\,3]} = \frac{\left(\sum_{i=1}^{6}\left(-\frac{i^2}{4}+1\right)\right)}{2}$$

$$\left(SR_{droite}\right)_{[0,\,3]} = -\frac{67}{8} \qquad \textbf{(12)}$$

```
> RiemannSum(f(x),x=0..3,
    method=right,
    partition = 6,
    output=plot,
    boxoptions = [filled = [color = turquoise, transparency = .5]]);
```




Une approximation de $\int_0^3 f(x)\,\mathrm{d}x$ par une somme droite de Rier

Résumons les sommes intégrales qui ont été obtenues:

$$SR_{gauche_{[0,\,3]}} = -\frac{31}{8} \qquad SR_{milieu_{[0,\,3]}} = -\frac{95}{16} \qquad SR_{droite_{[0,\,3]}} = -\frac{67}{8}$$

Avec un partition régulière de l'intervalle [0, 3] en six sous-intervalles, ces derniers résultats montrent que la manière de choisir le représentant dans chaque sous-intervalle a eu une grande incidence sur la valeur de ces sommes intégrales *SR*. Qu'en est-il de cette incidence sur chaque somme intégrale lorsque le nombre de sous-intervalles croît ?

Obtenons un tableau présentant la valeur de la somme intégrale gauche, milieu et droit lorsque le nombre de sous-intervalles créés par une partition régulière de l'intervalle [0, 3] croît.

```
> printf(`\n            Sommes intégrales avec partitions régulières de l'intervalle [0,3]\n`);
printf(`\n      |       n      |     SR[gauche]     |     SR[milieu]    |     SR[droite]    |\n`);
printf(`      ==============================================================================\n`);
 for k in [3,6,12,24,48,96,192,384,768,1536,3072,6144,12288,24576] do
 printf(`       |   %5d     |  %15.10f   |   %10.9f    | %15.10f |\n`,
 k,
 RiemannSum(f(x),x=0..3,method=left,partition = k,output=value),
 RiemannSum(f(x),x=0..3,method=midpoint,partition = k,output=value),
 RiemannSum(f(x),x=0..3,method=right,partition = k,output=value))
 od;
 k:='k':
```

```
          Sommes intégrales avec partitions régulières de l'intervalle [0,3]

      |       n      |     SR[gauche]     |     SR[milieu]     |     SR[droite]     |
      ===========================================================================
      |       3      |    -2.0000000000   |    -5.750000000    |  -11.0000000000    |
      |       6      |    -3.8750000000   |    -5.937500000    |   -8.3750000000    |
      |      12      |    -4.9062500000   |    -5.984375000    |   -7.1562500000    |
      |      24      |    -5.4453125000   |    -5.996093750    |   -6.5703125000    |
      |      48      |    -5.7207031250   |    -5.999023438    |   -6.2832031250    |
      |      96      |    -5.8598632810   |    -5.999755859    |   -6.1411132810    |
      |     192      |    -5.9298095700   |    -5.999938965    |   -6.0704345700    |
      |     384      |    -5.9648742680   |    -5.999984741    |   -6.0351867680    |
      |     768      |    -5.9824295040   |    -5.999996185    |   -6.0175857540    |
      |    1536      |    -5.9912128450   |    -5.999999046    |   -6.0087909700    |
      |    3072      |    -5.9956059460   |    -5.999999762    |   -6.0043950080    |
      |    6144      |    -5.9978028540   |    -5.999999940    |   -6.0021973850    |
      |   12288      |    -5.9989013970   |    -5.999999985    |   -6.0010986630    |
      |   24576      |    -5.9994506910   |    -5.999999996    |   -6.0005493240    |
```

Nous observons le tableau précédent que les sommes intégrales diffèrent sensiblement lorsque le nombre de sous-intervalles est assez petit mais, au fur et à mesure que le nombre de sous-intervalles croît, c'est-à-dire lorsque nous raffinons davantage la partition de l'intervalle [0,3], **chaque somme intégrale semble se rapprocher de $-6$.** Pour montrer qu'effectivement chaque somme intégrale $SR_{[0,\,3]} \to -6$ lorsque $n \to \infty$, obtenons dans un premier temps l'expression analytique de chacune de ces sommes intégrales

$SR_{[0,\,3]}$ avec une partition régulière de l'intervalle [0, 3] avec un nombre quelconque $n$ de sous-intervalles et, ensuite, évaluons la limite lorsque $n \to \infty$.

Commençons par développer analytiquement la somme intégrale $SR_{gauche_{[0,\,3]}}$.

```
> SR[gauche][[0,3]]:=RiemannSum(f(x),x=0..3,method=left,partition = n,
  output=sum);
  ``=expand(value(SR[gauche][[0,3]]));
```

$$\left(SR_{gauche}\right)_{[0,\,3]} := \frac{3\left(\sum_{i=0}^{n-1}\left(-\frac{9\,i^2}{n^2}+1\right)\right)}{n}$$
$$= -6 + \frac{27}{2\,n} - \frac{9}{2\,n^2} \tag{13}$$

Sous cette forme, il est facile de constater que $SR_{[0,\,3]} \to -6$ lorsque $n \to \infty$.

Évaluons ensuite la limite $\lim\limits_{n \to \infty} SR_{gauche_{[0,\,3]}}$.

```
> Limit('SR[gauche]'[[0,3]],n = infinity)=Limit((RiemannSum(f(x),x=0.
  .3,method=left,partition = n,output=sum)),n = infinity);
  ``=Limit(expand(value(RiemannSum(f(x),x=0..3,method=left,partition =
  n,output=sum))),n = infinity);
  ``=value(rhs(%));
```

$$\lim_{n \to \infty}\left(SR_{gauche}\right)_{[0,\,3]} = \lim_{n \to \infty} \frac{3\left(\sum_{i=0}^{n-1}\left(-\frac{9\,i^2}{n^2}+1\right)\right)}{n}$$
$$= \lim_{n \to \infty}\left(-6 + \frac{27}{2\,n} - \frac{9}{2\,n^2}\right)$$
$$= -6 \tag{14}$$

De même, développons la limite à l'infini de la somme intégrale formelle $SR_{milieu_{[0,\,3]}}$.

```
> SR[milieu][[0,3]]:=RiemannSum(f(x),x=0..3,method=midpoint,partition =
  n,output=sum);
  ``=expand(value(SR[milieu][[0,3]]));
```

$$\left(SR_{milieu}\right)_{[0,\,3]} := \frac{3\left(\sum_{i=0}^{n-1}\left(-\frac{9\left(i+\frac{1}{2}\right)^2}{n^2}+1\right)\right)}{n}$$
$$= \frac{9}{4\,n^2} - 6 \tag{15}$$

Évaluons maintenant la limite $\lim_{n\to\infty} SR_{milieu_{[0,\,3]}}$

```
> Limit('SR[milieu]'[[0,3]],n = infinity)=Limit((RiemannSum(f(x),x=0.
  .3,method=midpoint,partition = n,output=sum)),n = infinity);
  ``=Limit(expand(value(RiemannSum(f(x),x=0..3,method=midpoint,
  partition = n,output=sum))),n = infinity);
  ``=value(rhs(%));
```

$$\lim_{n\to\infty} \left(SR_{milieu}\right)_{[0,\,3]} = \lim_{n\to\infty} \frac{3\left(\sum_{i=0}^{n-1}\left(-\frac{9\left(i+\frac{1}{2}\right)^2}{n^2}+1\right)\right)}{n}$$
$$= \lim_{n\to\infty}\left(\frac{9}{4\,n^2}-6\right)$$
$$= -6 \qquad \textbf{(16)}$$

Terminons avec le calcul de la limite $\lim_{n\to\infty} SR_{droite_{[0,\,3]}}$.

```
> SR[droite][[0,3]]:=RiemannSum(f(x),x=0..3,method=right,partition = n,
  output=sum);
  ``=expand(value(SR[droite][[0,3]]));
```

$$\left(SR_{droite}\right)_{[0,\,3]} := \frac{3\left(\sum_{i=1}^{n}\left(-\frac{9\,i^2}{n^2}+1\right)\right)}{n}$$
$$= -6-\frac{27}{2\,n}-\frac{9}{2\,n^2} \qquad \textbf{(17)}$$

Évaluons maintenant la limite $\lim_{n\to\infty} SR_{droit_{[0,\,3]}}$

```
> Limit('SR[droite]'[[0,3]],n = infinity)=Limit((RiemannSum(f(x),x=0.
  .3,method=right,partition = n,output=sum)),n = infinity);
  ``=Limit(expand(value(RiemannSum(f(x),x=0..3,method=right,partition =
  n,output=sum))),n = infinity);
  ``=value(rhs(%));
```

$$\lim_{n\to\infty} \left(SR_{droite}\right)_{[0,\,3]} = \lim_{n\to\infty} \frac{3\left(\sum_{i=1}^{n}\left(-\frac{9\,i^2}{n^2}+1\right)\right)}{n}$$
$$= \lim_{n\to\infty}\left(-6-\frac{27}{2\,n}-\frac{9}{2\,n^2}\right)$$
$$= -6 \qquad \textbf{(18)}$$

Notons que pour chaque formulations formelles *SR* simplifiées précédentes, il n'était nullement nécessaire ici, de recourir à Maple, pour évaluer la limite à l'infini.

Pour interpréter graphiquement ces limites, observons la progression de ces sommes intégrales dans leur représentation graphique respective avec des partitions régulières de 6, 12, 24 et 48 sous-intervalles. Les représentations ci-dessous ont tous été obtenus en spécifiant l'argument *caption* = ``. Observons alors que plus la partition se raffine, davantage les représentations graphiques de chaque somme intégrale se confondent.

Partition régulière avec 6 sous-intervalles




Partition régulière avec 12 sous-intervalles




Partition régulière avec 24 sous-intervalles




Partition régulière avec 48 sous-intervalles




Lorsque la limite $\lim_{\max \Delta x_i \to 0} \left( \sum_{i=1}^{n} f\left(c_i\right) \Delta x_i \right)$ existe pour une fonction *f* définie sur un intervalle [*a*,*b*], nous disons que la fonction *f* est intégrable au sens de Riemann et que la valeur de cette limite est appelée l'intégrale définie de la fonction *f* sur [*a*,*b*] et sera notée $\int_a^b f(x)\, \mathrm{d}x$.

$$\int_a^b f(x)\, \mathrm{d}x = \lim_{\max \Delta x_i \to 0} \left( \sum_{i=1}^{n} f\left(c_i\right) \Delta x_i \right), \qquad c_i \in \left[x_{i-1}, x_i\right]$$

Il a été montrer que, si la limite

$$\lim_{\max \Delta x_i \to 0} \left( \sum_{i=1}^{n} f\left(c_i\right) \Delta x_i \right)$$ existe, elle est indépendante du représentant choisi

$c_i$. Alors, il aurait suffit de calculer $\lim_{n \to \infty} SR_{gauche_{[0,\,3]}}$ pour conclure que $\int_0^3 \left(1 - x^2\right) \mathrm{d}x = -6$. Par contre, cela nous a permis de constater que nous arrivons à la même valeur $-6$ avec $SR_{milieu_{[0,\,3]}}$ et $SR_{droite_{[0,\,3]}}$.

Ce que nous obtenons lorsque nous calculons une intégrale définie est donc un nombre réel et non pas une fonction. Pour une fonction f donnée, la valeur de l'intégrale définie, bien que dépendante de l'intervalle [*a*,*b*], ne dépend pas de la variable *x*. Dans la notation de l'intégrale définie, la lettre *x* joue un rôle semblable que la lettre *i* dans la notation $\Sigma$.

$$\int_a^b f(x)\, \mathrm{d}x \in \mathbb{R}.$$

Ce nombre réel existe-t-il toujours quelque soit la fonction *f* et quelque soit l'intervalle [*a*,*b*] ? Autrement dit, une fonction *f* est-elle toujours intégrable sur un intervalle [*a*,*b*] ? Pas toujours, mais, il a été prouvé que, si *f* est une fonction continue sur un intervalle [*a*,*b*], alors la fonction *f* est intégrable sur [*a*,*b*].

# Macro-commandes plotriemann et coloriage

La macro-commande ***plotriemann*** de l'extension *riemann* permet d'aller beaucoup plus loin dans le calcul et dans la représentation graphique des sommes intégrales au sens de Riemann:

- en précisant une partition pouvant être *régulière* ou *explicitement non-régulière* ou *générer aléatoirement*.
- en précisant explicitement un choix arbitraire d'une liste de représentants ou en générant automatiquement un choix aléatoire de représentants dans les sous-intervalles créés par la partition. On pourra, bien sûr, préciser également les choix systématiques habituels « gauche », « milieu » et « droit »
- en obtenant les sommes inférieure et supérieure de Riemann ainsi que leur représentation graphique.

La macro-commande

```
plotriemann(Lieu,Intervalle,Partition,Représentant,
Calcul)
```

possède cinq arguments obligatoires permettant d'obtenir la valeur ou une approximation d'une somme intégrale au sens de Riemann et de la représenter graphiquement.

Description des arguments:

- **`Lieu`**: La courbe à tracer doit être formulée de la façon suivante: [$x$, f($x$), $x = a..b$].
- **`Intervalle`**: L'intervalle à partitionner doit être énoncé dans la forme $x = c\,..\,d$. L'intervalle [$c$,$d$] étant bien sûr un sous-ensemble de l'intervalle [$a$,$b$].
- **`Partition`**: La partition $P$ de l'intervalle [$c$,$d$] sera régulière si on formule le troisième argument comme $n = k$, $k > 0$. Sinon, le troisième argument doit être une liste de la forme [ $x_0, x_1, x_2, x_3, \ldots, x_{n-1}, x_n$] constituant une partition $P$ de l'intervalle [$c$,$d$]. Cette liste pouvant être explicite ou être générée aléatoirement.

- **Représentant**: Cet argument précise la manière dont le représentant $c_i$ sera choisi dans chacun des sous-intervalles créés par la partition $P$. La valeur de l'argument Représentant peut être:
  - gauche pour que le représentant soit choisi systématiquement comme la borne inférieure de chaque sous-intervalle;
  - milieu pour que le représentant soit choisi systématiquement comme le milieu de chaque sous-intervalle;
  - droit pour que le représentant soit choisi systématiquement comme la borne supérieure de chaque sous-intervalle;
  - une liste de représentants $[c_1, c_2, c_3, ..., c_{i-1}, c_i, c_{i+1}, ..., c_{n-1}, c_n]$ choisis explicitement par l'utilisateur;
  - aléatoire pour que le représentant soit choisi aléatoirement dans chaque sous-intervalle;
  - infsup pour que les représentants soit choisis respectivement comme l'abscisse du minimum et comme l'abscisse du maximum de la fonction dans chaque sous-intervalle.
- **Calcul**: Cet argument précise si l'approximation numérique de la somme intégrale sera donnée ou non. La forme de l'argument Calcul doit être:
  - calcul=oui pour obtenir une approximation numérique de la somme intégrale et sa représentation graphique;
  - calcul=non pour obtenir seulement la représentation graphique de la somme intégrale;
  - calcul=seul pour obtenir seulement une approximation numérique de la somme intégrale sans la représentation graphique.

La macro-commande

**coloriage(Lieu,Intervalle)**

possède deux arguments permettant de tracer la courbe d'équation $y = f(x)$ et de colorer la région comprise entre la courbe et l'axe des $x$. La couleur turquoise sera utilisée pour la région au-dessus de l'axe des $x$ et la couleur rose pour celle au-dessous de l'axe des $x$.

Description des arguments:

- **Lieu**: La courbe à tracer doit être formulée de la façon suivante: $[x, f(x), x = a..b]$
- **Intervalle**: L'intervalle correspondant à la région à colorer doit être énoncé dans la forme $x = c\,..\,d$. L'intervalle $[c,d]$ étant bien sûr un sous-ensemble de l'intervalle $[a,b]$.

Les macro-commandes ***coloriage*** et ***plotriemann*** font parties de la bibliothèque ***riemann*** que j'ai écris pour mes élèves en 2002. Pour la séance Maple qui est en cours, vous devrez donc indiquer à Maple où aller lire les instructions composant les macro-commandes de cette bibliothèque personnelle. Vous devrez donc taper dans une zone de requêtes la requête suivante:

```
with(riemann);
```

Mais, préalablement, vous devrez indiquer à Maple le chemin à suivre pour accéder aux macro-commandes de cette bibliothèque, c'est-à dire accéder au fichier contenant le code de ces macro-commandes. Il faut donc informer Maple du chemin menant au dossier contenant le fichier dans lequel il trouvera cette bibliothèque. Pour disposer des macro-commandes sur votre ordinateur, il vous faudra procéder à l'installation de deux fichiers « riemann.ind » et « riemann.lib » sur votre disque rigide: celui de votre ordinateur personnel et non pas celui du poste sur lequel vous travaillez au collège.

À partir de mon site internet, ........................., cliquer sur l'onglet «NYB», localiser et télécharger les fichiers « riemann.ind » et « riemann.lib » et les déposer dans un sous-dossier de votre ordinateur. Je vous suggère de nommer ce sous-dossier « Bibliothèque Riemann » et un bon choix d'emplacement pour ce dossier serait de le créer dans le dossier *Users* de votre installation Maple.

Ensuite, la façon d'informer Maple du chemin dont il est question consiste **à ajouter** à la variable d'environnement `libname` ce chemin. Mais d'abord, l'énoncé du chemin doit être une **chaîne de caractères** de la forme

"Nom_du_volume:\\Répertoire_1\\Répertoire_2\\....../Répertoire_contenant_le_fichier"

ATTENTION: La variable `libname` possède déjà une assignation qui a été faite au moment de l'installation du logiciel sur votre ordinateur, soit l'assignation du chemin menant à la bibliothèque de base **maple.lib** contenu dans le dossier lib. Voyez:

```
> libname;
```

"C:\Program Files\Maple 2020\lib" **(19)**

Il ne faut surtout pas écraser le contenu de la variable `libname` par une nouvelle assignation, Maple ne saurait plus comment localiser les macro-commandes de la bibliothèque de base maple.lib. La façon dont il faut s'y prendre est la suivante:

*libname := nouveau_chemin, libname*;

Par exemple, l'assignation suivante est celle que je dois faire son mon ordinateur personnel.

```
> libname:="C:\\Users\\plant\\Desktop\\Éléments site Maple au
  cégep\\Downloads\\Documents Maple\\NYB/Bibliothèque riemann",libname;
```

*libname* := **(20)**

"C:\Users\plant\Desktop\Éléments site Maple au cégep\Downloads\Documents Maple\NYB/Bibliothèque riemann", "C:\Program Files\Maple 2020\lib"

On remarquera le dernier élément dans le résultat de cette assignation: c'est le chemin menant à la bibliothèque de base, c'est-à-dire que c'est le contenu qu'avait la variable *libname*. Cette façon de faire pour ajouter (concaténer) un nouveau terme à une séquence est valable quelque soit la variable en cause. Dans ce cas-ci, la variable en cause est la variable *libname*.

Maintenant, nous pouvons indiquer à Maple la manière d'accéder aux macro-commandes *coloriage* et *plotriemann* de la bibliothèque ***riemann***.

```
> with(riemann,coloriage,plotriemann);
```

$[coloriage, plotriemann]$ **(21)**

# Exemples

## Exemple 1

Insistons sur le fait que le concept de somme intégrale est un calcul mathématique dont le résultat est un nombre réel.

Soit la fonction *f* définie par $f(x) = \sin(x)$ sur l'intervalle $[-\pi, \pi]$. Calculons $\sum_{i=1}^{15} \sin(c_i)\, \Delta x_i$ en appliquant la définition d'une somme intégrale droit avec une partition régulière de l'intervalle $[-\pi, \pi]$ en

15 sous-intervalles et en choisissant systématiquement le borne supérieure de chaque sous-intervalle de la partition.

Créons la fonction f.

```
> f:=x->sin(x);
```

$$f := x \mapsto \sin(x) \tag{22}$$

Obtenons l'expression analytique de $SR_{droite_{[-\pi,\,\pi]}}$ en utilisant la macro-commande *RiemannSum*.

```
> SR[droit][[-Pi,Pi]]:=RiemannSum(f(x),x=-Pi..Pi,method=right,
  partition = 15,output=sum);
```

$$\left(SR_{droit}\right)_{[-\pi,\,\pi]} := \frac{2\pi\left(\sum_{i=1}^{15}\left(-\sin\left(\frac{2 i \pi}{15}\right)\right)\right)}{15} \tag{23}$$

```
> SR[droit][[-Pi,Pi]]=value(SR[droit][[-Pi,Pi]]);
```

$$\frac{2\pi\left(\sum_{i=1}^{15}\left(-\sin\left(\frac{2 i \pi}{15}\right)\right)\right)}{15} = 0 \tag{24}$$

Représentons graphiquement, avec la macro-commande ***plotriemann***, cette somme intégrale droit pour mieux " voir" ce qui expliquerait le résultat nul de cette somme intégrale.

```
> plotriemann([x,f(x),x=-Pi..Pi],x=-Pi..Pi,n=15,droit,calcul=non,
  title="SR[droit] avec n="||15,titlefont=[COURIER,8]);
```




Dans la définition de la somme intégrale $\sum_{i=1}^{15} \mathrm{f}\left(c_i\right) \Delta x_i$, $\Delta x_i$ étant toujours positif, le signe du produit $f\left(c_i\right) \Delta x_i$ sera celui du signe de $f\left(c_i\right)$. Alors,

– lorsque $x \in [-\pi, 0]$, $f\left(c_i\right) < 0$ et, dans ce cas, chaque terme $f\left(c_i\right) \Delta x_i < 0$ (négatif) est représenté par un rectangle de couleur rose

– lorsque $x \in [0, \pi]$, $f\left(c_i\right) > 0$ et, dans ce cas, chaque terme $f\left(c_i\right) \Delta x_i > 0$ (positif) est représenté par un rectangle de couleur turquoise

En résumé, chaque rectangle rose est donc associé à une valeur négative tandis que chaque rectangle turquoise est associé à une valeur positive. Dans ce cas-ci, étant donné la symétrie de la situation, cela explique très bien pourquoi la somme $\sum_{i=1}^{15}\left(-\sin\left(\frac{2\,i\,\pi}{15}\right)\right)$ donne 0 sans même qu'il ne soit nécessaire de calculer explicitement cette somme

$$\sum_{i=1}^{15}\left(-\sin\left(\frac{2\,i\,\pi}{15}\right)\right) =$$

$$(-\sin)\left(\frac{2\pi}{15}\right) + (-\sin)\left(\frac{4\pi}{15}\right) + (-\sin)\left(\frac{2\pi}{5}\right) + (-\sin)\left(\frac{8\pi}{15}\right) + (-\sin)\left(\frac{2\pi}{3}\right) + (-\sin)\left(\frac{4\pi}{5}\right) + (-\sin)\left(\frac{14\pi}{15}\right) + (-\sin)\left(\frac{16\pi}{15}\right) + (-\sin)\left(\frac{6\pi}{5}\right) + (-\sin)\left(\frac{4\pi}{3}\right) + (-\sin)\left(\frac{22\pi}{15}\right) + (-\sin)\left(\frac{8\pi}{5}\right) + (-\sin)\left(\frac{26\pi}{15}\right) + (-\sin)\left(\frac{28\pi}{15}\right) + (-\sin)(2\pi)$$

Obtenons quand même la valeur numérique de chaque terme pour clairement mettre en évidence la symétrie de la situation.

```
> Sum(-sin(2/15*i*Pi),i = 1 .. 15)=add('-sin'(2/15*i*Pi),i = 1 ..
  15);
```

$$\sum_{i=1}^{15}\left(-\sin\left(\frac{2\,i\,\pi}{15}\right)\right) = (-\sin)\left(\frac{2\pi}{15}\right) + (-\sin)\left(\frac{4\pi}{15}\right) + (-\sin)\left(\frac{2\pi}{5}\right) + (-\sin)\left(\frac{8\pi}{15}\right) + (-\sin)\left(\frac{2\pi}{3}\right) + (-\sin)\left(\frac{4\pi}{5}\right) + (-\sin)\left(\frac{14\pi}{15}\right) + (-\sin)\left(\frac{16\pi}{15}\right) + (-\sin)\left(\frac{6\pi}{5}\right) + (-\sin)\left(\frac{4\pi}{3}\right) + (-\sin)\left(\frac{22\pi}{15}\right) + (-\sin)\left(\frac{8\pi}{5}\right) + (-\sin)\left(\frac{26\pi}{15}\right) + (-\sin)\left(\frac{28\pi}{15}\right) + (-\sin)(2\pi) \qquad \textbf{(25)}$$

Alors,

```
> Termes:=[seq((-sin(2/15*i*Pi)),i=1..15)];
```

$$Termes := \left[-\sin\left(\frac{2\pi}{15}\right), -\sin\left(\frac{4\pi}{15}\right), -\sin\left(\frac{2\pi}{5}\right), -\sin\left(\frac{7\pi}{15}\right), -\frac{\sqrt{3}}{2}, -\sin\left(\frac{\pi}{5}\right), -\sin\left(\frac{\pi}{15}\right), \sin\left(\frac{\pi}{15}\right), \sin\left(\frac{\pi}{5}\right), \frac{\sqrt{3}}{2}, \sin\left(\frac{7\pi}{15}\right), \sin\left(\frac{2\pi}{5}\right), \sin\left(\frac{4\pi}{15}\right), \sin\left(\frac{2\pi}{15}\right), 0\right] \qquad \textbf{(26)}$$

```
> evalf(Termes);
```

[ −0.4067366430, −0.7431448256, −0.9510565165, −0.9945218954, −0.8660254040, −0.5877852524, −0.2079116909, 0.2079116909, 0.5877852524, 0.8660254040, 0.9945218954, 0.9510565165, 0.7431448256, 0.4067366430, 0. ] **(27)**

En ré-écrivant l'addition de la manière suivante, il est clair que la somme est nulle.

$$\sum_{i=1}^{15}\left(-\sin\left(\frac{2\,i\,\pi}{15}\right)\right) = (\ -0.4067366430 + 0.4067366430) + (-0.7431448256 + 0.7431448256) + \ldots + (-0.2079116909 + 0.2079116909) + 0$$

## Exemple 2

Obtenons la valeur de $\int_{-3}^{5} \left(x^2 - x - 7\right)\, dx$ en appliquant la définition de l'intégrale définie. Développons cette intégrale définie avec une somme intégrale droite.

Créons la fonction *f* et obtenons son tracé sur l'intervalle [-4, 5].

```
> f:=x->x^2-x-7;
  plot([x,f(x),x=-4..5]);
```

$$f := x \mapsto x^2 - x - 7$$




Obtenons l'expression formelle de $SR_{droite_{[-3,\,5]}}$ en utilisant la macro-commande *RiemannSum*.

```
> SR[droite][[-3,5]]:=RiemannSum(f(x),x=-3..5,method=right,partition
  = n,output=sum);
  ``=expand(value(SR[droite][[-3,5]]));
```

$$\left(SR_{droite}\right)_{[-3,\,5]} := \frac{8\left(\sum_{i=1}^{n}\left(\left(-3 + \frac{8\,i}{n}\right)^2 - 4 - \frac{8\,i}{n}\right)\right)}{n}$$

$$= -\frac{40}{3} + \frac{32}{n} + \frac{256}{3\,n^2} \qquad \textbf{(28)}$$

Et alors, par définition, nous avons que $\int_{-3}^{5} \left(x^2 - x - 7\right)\, dx = \lim_{n\to\infty} SR_{droite_{[-3,\,5]}}$

```
> Int(f(x),x=-3..5)=Limit('SR[droite]'[[-3,5]],n = infinity);
```

```
``=Limit(expand(value(SR[droite][[-3,5]])),n = infinity);
value(%);
```

$$\int_{-3}^{5} \left(x^2 - x - 7\right)\, dx = \lim_{n \to \infty} \left(SR_{droite}\right)_{[-3,\,5]}$$
$$= \lim_{n \to \infty} \left(-\frac{40}{3} + \frac{32}{n} + \frac{256}{3\,n^2}\right)$$
$$= -\frac{40}{3} \quad \textbf{(29)}$$

Notons que $\int_{-3}^{5} \left(x^2 - x - 7\right)\, dx < 0$.

Représentons graphiquement cette intégrale définie.

```
> coloriage([x,f(x),x=-4..5],x=-3..5);
```



Afin de visualiser que le coloriage précédent est en quelque sorte le coloriage limite que nous pourrions réaliser avec des sommes intégrales correspondant à des partitions de plus en plus fines, représentons graphiquement différentes sommes intégrales droites avec des partitions régulières de l'intervalle [-3, 5] en 15, 25 et 35 sous-intervalles .

```
> SR[d]:=seq(plotriemann([x,f(x),x=-4..5],x=-3..5,n=k,droit,calcul=
non,title="SR[droit] où n="||k,titlefont=[COURIER,8],axesfont=
[TIMES,ROMAN,8]),k=[15,25,35]):
plots[display](array(1..3,[SR[d][1],SR[d][2],SR[d][3]]),
xtickmarks=6);
```




Au besoin, sélectionner le graphique précédent et étirer-le pour mieux voir les différentes représentations.

## Exemple 3

$\int_a^b f(x)\,dx$ est un calcul mathématique qui peut être interpréter géométriquement comme un calcul d' "aire algébrique" de la surface comprise entre la courbe et l'axe des $x$ sur l'intervalle [a,b].

Dans la définition de la somme intégrale $\sum_{i=1}^{n} f(c_i)\,\Delta x_i$, $\Delta x_i$ étant toujours positif, le signe du produit $f(c_i)\,\Delta x_i$ sera celui du signe de $f(c_i)$. Le produit $f(c_i)\,\Delta x_i$ est interprété comme un calcul d'aire *algébrique* du rectangle de largeur $\Delta x_i$

– au-dessus de l'axe des $x$ si $f(c_i) > 0$ et dans ce cas $f(c_i)\,\Delta x_i > 0$
– au-dessous de l'axe des $x$ si $f(c_i) < 0$ et dans ce cas $f(c_i)\,\Delta x_i < 0$

Ainsi, lorsque $f(x) > 0$ sur un intervalle [*c*,*d*], la somme intégrale $\sum_{i=1}^{n} f(c_i)\,\Delta x_i$ associée à une partition de l'intervalle [*c*,*d*] correspond à la somme des aires des rectangles ***au-dessus de l'axe des* x** et la valeur de la somme intégrale sera positive. Ainsi, à la limite, le calcul de $\int_c^d f(x)\,dx$ donne directement la mesure de l'aire de la surface comprise entre la courbe *f* et l'axe des *x*.

Par contre, lorsque $f(x)$ est négative sur un intervalle [*c*,*d*], la somme intégrale $\sum_{i=1}^{n} f(c_i)\,\Delta x_i$, quant à elle, ne correspond pas à la somme des aires des rectangles ***au-dessous de l'axe des* x** car la dimension $f(c_i)$ de chaque rectangle est négative. Dans ce cas, c'est plutôt le calcul $\sum_{i=1}^{n}\left(-f(c_i)\,\Delta x_i\right)$ qui donnera l'aire de la surface comprise entre la courbe et l'axe des *x*. Par une propriété du symbole de sommation,

$$\sum_{i=1}^{n}\left(-f(c_i)\,\Delta x_i\right) = -\left(\sum_{i=1}^{n} f(c_i)\,\Delta x_i\right)$$

Ainsi, à la limite, c'est le calcul de $-\left(\int_c^d f(x)\,dx\right)$ qui donnera la mesure de l'aire de la surface comprise entre la courbe f et l'axe des $x$ lorsque la surface est au-dessous de l'axe des $x$.

L'exemple suivant devrait facilement vous convaincre que l'intégrale définie $\int_a^b f(x)\,dx$ est un calcul algébrique pouvant être interprété comme celui d'un calcul d'aire ***sous certaines conditions***.

Pour mieux illustrer et donc pour mieux documenter les prochains développements, nous allons utiliser la macro-commande ***coloriage*** de l'extension ***riemann***.

Soit la fonction $f$ définie par $f(x) = -x^3 + 2x^2 + 5x$. Calculons l'aire (géométrique) de la région comprise entre la courbe et l'axe des $x$.

```
> f:=x->-x^3+2*x^2+5*x;
  plot([x,f(x),x=-2..4],color=navy);
```

$$f := x \mapsto -x^3 + 2\cdot x^2 + 5\cdot x$$




Afin de colorier exactement la région comprise entre la courbe et l'axe des $x$, nous devons trouver les zéros de la fonction $f$. Comme fonction polynomiale de degré 3, le tracé précédent nous montre bien les trois zéros réels de la fonction $f$. Puisque le terme constant de $f(x)$ est nul, 0 est un zéro de la fonction. Ensuite, obtenons directement avec Maple le plus petit et le plus grand zéro de la fonction $f$.

```
> a:=min(solve(f(x)=0,x));
  b:=max(solve(f(x)=0,x));
```

$$a := 1 - \sqrt{6}$$

$$b := 1 + \sqrt{6} \tag{30}$$

Colorons maintenant la région dont nous allons calculer l'aire. La macro-commande ***coloriage*** de l'extension ***riemann*** colore la région comprise entre l'axe des $x$ et la courbe.

```
> coloriage([x,f(x),x=-2..4],x=a..b);
```




Le calcul de l'aire doit être réalisé au moyen du calcul de deux intégrales définies:

– par $A_1 = -\left(\int_{1-\sqrt{6}}^{0} \left(-x^3+2x^2+5x\right) dx\right)$ pour la région de couleur « rose » (région au-dessous de l'axe des $x$)

– par $A_2 = \int_{0}^{1+\sqrt{6}} \left(-x^3+2x^2+5x\right) dx$ pour la région de couleur « turquoise » (région au-dessus de l'axe des $x$)

Pour obtenir $\int_{1-\sqrt{6}}^{0} \left(-x^3+2x^2+5x\right) dx$, obtenons d'abord l'expression formelle de la somme intégrale $SR_{gauche_{\left[1-\sqrt{6},\,0\right]}}$.

```
> SR[gauche][[a,0]]:=RiemannSum(f(x),x=a..0,method=left,partition =
  n,output=sum);
  ``=expand(value(SR[gauche][[a,0]]));
```

$$\left(SR_{gauche}\right)_{\left[1-\sqrt{6},\,0\right]} := \frac{1}{n}\left(\left(-1+\sqrt{6}\right)\left(\sum_{i=0}^{n-1}\left(-\left(1-\sqrt{6}+\frac{i\left(-1+\sqrt{6}\right)}{n}\right)^3+2\left(1-\sqrt{6}+\frac{i\left(-1+\sqrt{6}\right)}{n}\right)^2+5-5\sqrt{6}+\frac{5\,i\left(-1+\sqrt{6}\right)}{n}\right)\right)\right)$$

$$= 4\sqrt{6}-\frac{143}{12}-\frac{4\sqrt{6}}{n^2}+\frac{143}{12\,n^2} \qquad \textbf{(31)}$$

Reste ensuite à obtenir la limite à l'infinie de cette somme.

```
> Int(f(x),x=a..0)=Limit('SR[gauche]'[[a,0]],n = infinity);
  ``=Limit(expand(value(SR[gauche][[a,0]])),n = infinity);
```

```
``=value(rhs(%));
```

$$\int_{1-\sqrt{6}}^{0} \left(-x^3+2x^2+5x\right) dx = \lim_{n\to\infty} \left(SR_{gauche}\right)_{\left[1-\sqrt{6},\,0\right]}$$

$$= \lim_{n\to\infty} \left(4\sqrt{6} - \frac{143}{12} - \frac{4\sqrt{6}}{n^2} + \frac{143}{12\,n^2}\right)$$

$$= 4\sqrt{6} - \frac{143}{12} \quad \textbf{(32)}$$

Donc, l'aire de la région de couleur « rose » est $A_1 = -\left(\int_{1-\sqrt{6}}^{0} \left(-x^3+2x^2+5x\right) dx\right) = \left(-\frac{143}{12}+4\sqrt{6}\right) u^2$:

```
> A[1]:=-(-143/12+4*sqrt(6))*(u^2);
```

$$A_1 := -\left(4\sqrt{6} - \frac{143}{12}\right) u^2 \quad \textbf{(33)}$$

Obtenons ensuite $A_2 = \int_{0}^{1+\sqrt{6}} \left(-x^3+2x^2+5x\right) dx$.

```
> SR[gauche][[0,b]]:=RiemannSum(f(x),x=0..b,method=left,partition =
  n,output=sum);
  ``=expand(value(SR[gauche][[0,b]]));
```

$$\left(SR_{gauche}\right)_{\left[0,\,1+\sqrt{6}\right]} :=$$

$$\frac{\left(1+\sqrt{6}\right)\left(\sum_{i=0}^{n-1}\left(-\frac{i^3\left(1+\sqrt{6}\right)^3}{n^3}+\frac{2\,i^2\left(1+\sqrt{6}\right)^2}{n^2}+\frac{5\,i\left(1+\sqrt{6}\right)}{n}\right)\right)}{n}$$

$$= \frac{143}{12} - \frac{143}{12\,n^2} + 4\sqrt{6} - \frac{4\sqrt{6}}{n^2} \quad \textbf{(34)}$$

```
> Int(f(x),x=0..b)=Limit('SR[gauche]'[[0,b]],n = infinity);
  ``=Limit(expand(value(SR[gauche][[0,b]])),n = infinity);
  ``=value(rhs(%));
```

$$\int_{0}^{1+\sqrt{6}} \left(-x^3+2x^2+5x\right) dx = \lim_{n\to\infty} \left(SR_{gauche}\right)_{\left[0,\,1+\sqrt{6}\right]}$$

$$= \lim_{n\to\infty} \left(\frac{143}{12} - \frac{143}{12\,n^2} + 4\sqrt{6} - \frac{4\sqrt{6}}{n^2}\right)$$

$$= \frac{143}{12} + 4\sqrt{6} \quad \textbf{(35)}$$

Donc, l'aire de la région de couleur turquoise est

```
> A[2]:=(143/12+4*sqrt(6))*(u^2);
```

$$A_2 := \left(\frac{143}{12} + 4\sqrt{6}\right) u^2 \qquad \textbf{(36)}$$

L'aire cherchée est alors la somme $A_1 + A_2$.

```
> Aire:=A[1]+A[2];
```

$$Aire := -\left(4\sqrt{6} - \frac{143}{12}\right) u^2 + \left(\frac{143}{12} + 4\sqrt{6}\right) u^2 \qquad \textbf{(37)}$$

```
> ``=radnormal(Aire);
```

$$= \frac{143\, u^2}{6} \qquad \textbf{(38)}$$

La région colorée suivante

```
> coloriage([x,f(x),x=-2..4],x=a..b);
```




est la situation limite d'un coloriage de la représentation graphique d'une somme intégrale correspondant, par exemple, à une partition régulière de l'intervalle $\left[1 - \sqrt{6}, 1 + \sqrt{6}\right]$ dont le représentant $c_i$ est choisi systématiquement comme la borne inférieure de chaque sous-intervalle de la partition.

```
> plotriemann([x,f(x),x=-2..4],x=a..b,n=36,gauche,calcul=non);
```




Notons finalement que le calcul de $\int_{1-\sqrt{6}}^{1+\sqrt{6}} \left(-x^3+2x^2+5x\right) \mathrm{d}x$ comme la somme de $\int_{1-\sqrt{6}}^{0} \left(-x^3+2x^2+5x\right) \mathrm{d}x$ et de $\int_{0}^{1+\sqrt{6}} \left(-x^3+2x^2+5x\right) \mathrm{d}x$, est bien que conforme à la propriété suivante de l'intégrale définie.

$$\int_a^b \mathrm{f}(x)\ \mathrm{d}x = \int_a^c \mathrm{f}(x)\ \mathrm{d}x + \int_c^b \mathrm{f}(x)\ \mathrm{d}x, \quad a, b \text{ et } c \in \mathbb{R}$$

ne répond pas à la question du calcul d'aire. En effet,

```
> 'SR[gauche]'[[a,b]]='SR[gauche]'[[a,0]]+'SR[gauche]'[[0,b]];
  ``=value(SR[gauche][[a,0]]+SR[gauche][[0,b]]);
  radnormal(%);
```

$$\left(SR_{gauche}\right)_{\left[1-\sqrt{6},\, 1+\sqrt{6}\right]} = \left(SR_{gauche}\right)_{\left[1-\sqrt{6},\, 0\right]} + \left(SR_{gauche}\right)_{\left[0,\, 1+\sqrt{6}\right]}$$

$$= \frac{\left(-1+\sqrt{6}\right)\left(-\frac{19\sqrt{6}\ n}{12} + \frac{29\ n}{12} + \frac{19\sqrt{6}}{12\ n} - \frac{29}{12\ n}\right)}{n}$$

$$+ \frac{\left(1+\sqrt{6}\right)\left(\frac{29\ n}{12} - \frac{29}{12\ n} + \frac{19\sqrt{6}\ n}{12} - \frac{19\sqrt{6}}{12\ n}\right)}{n}$$

$$= \frac{8\sqrt{6}\left(n^2-1\right)}{n^2} \qquad \textbf{(39)}$$

D'où

```
> Int(-x^3+2*x^2+5*x,x = 1-sqrt(6) .. 1+sqrt(6))=Limit('SR[gauche]'[
  [a,b]],n=infinity);
  ``=Limit(8*6^(1/2)*(n^2-1)/n^2,n=infinity);
  value(%);
```

$$\int_{1-\sqrt{6}}^{1+\sqrt{6}} \left(-x^3 + 2x^2 + 5x\right) dx = \lim_{n \to \infty} \left(SR_{gauche}\right)_{\left[1-\sqrt{6},\, 1+\sqrt{6}\right]}$$

$$= \lim_{n \to \infty} \frac{8\sqrt{6}\left(n^2 - 1\right)}{n^2}$$

$$= 8\sqrt{6} \qquad \textbf{(40)}$$

Il est évident que $\frac{143}{6} \neq 8\sqrt{6}$.

## Exemple 4

Cette exemple illustrera le calcul d'une intégrale définie dans le cas d'une fonction continue par morceaux. **Il ne s'agit pas ici d'un calcul d'aire**.

Soit la fonction *f* ci-dessous définie sur l'intervalle [-1, 3]

```
> f:=x->piecewise(x<0,x^3+1,x>=0 and x<2,x^2-1, x>=2, -sqrt(x-2)):
  'f'(x)=f(x);
  plot([x,f(x),x=-1..3],color=navy,discont=true);
```

$$f(x) = \begin{cases} x^3 + 1 & x < 0 \\ x^2 - 1 & 0 \leq x < 2 \\ -\sqrt{-2 + x} & 2 \leq x \end{cases}$$




Le calcul de $\int_{-1}^{3} f(x)\, dx$ **doit être** développé par la somme de

$\int_{-1}^{0} \left(x^3 + 1\right) dx + \int_{0}^{2} \left(x^2 - 1\right) dx + \int_{2}^{3} \left(-\sqrt{x-2}\right) dx$. Représentons graphiquement cette somme.

```
> coloriage([x,f(x),x=-1..3],x=-1..3);
```




Pour chaque intégrale définie, appliquons la définition de l'intégrale définie avec une partition régulière et en utilisant une somme intégrale milieu.

```
> SR[milieu][[-1,0]]:=RiemannSum(x^3+1,x=-1..0,method=midpoint,
  partition = n,output=sum);
  ``=expand(value(SR[milieu][[-1,0]]));
```

$$\left(SR_{milieu}\right)_{[-1,\,0]} := \frac{\displaystyle\sum_{i=0}^{n-1}\left(\left(-1+\frac{i+\frac{1}{2}}{n}\right)^3+1\right)}{n}$$

$$= \frac{1}{8\,n^2}+\frac{3}{4} \tag{41}$$

```
> Int(x^3+1,x=-1..0)=Limit('SR[milieu]'[[-1,0]],n = infinity);
  S[1]:=Limit(expand(value(SR[milieu][[-1,0]])),n = infinity);
```

$$\int_{-1}^{0}\left(x^3+1\right)\,dx = \lim_{n\to\infty}\left(SR_{milieu}\right)_{[-1,\,0]}$$

$$S_1 := \lim_{n\to\infty}\left(\frac{1}{8\,n^2}+\frac{3}{4}\right) \tag{42}$$

```
> SR[milieu][[0,2]]:=RiemannSum(x^2-1,x=0..2,method=midpoint,
  partition = n,output=sum);
  ``=expand(value(SR[milieu][[0,2]]));
  Int(x^2-1,x=0..2)=Limit('SR[milieu]'[[0,2]],n = infinity);
  S[2]:=Limit(expand(value(SR[milieu][[0,2]])),n = infinity);
```

$$\left(SR_{milieu}\right)_{[0,\,2]} := \frac{2\left(\displaystyle\sum_{i=0}^{n-1}\left(\frac{4\left(i+\frac{1}{2}\right)^2}{n^2}-1\right)\right)}{n}$$

$$= -\frac{2}{3\,n^2} + \frac{2}{3}$$

$$\int_0^2 \left(x^2 - 1\right) \, dx = \lim_{n \to \infty} \left(SR_{milieu}\right)_{[0,\,2]}$$

$$S_2 := \lim_{n \to \infty} \left(-\frac{2}{3\,n^2} + \frac{2}{3}\right) \tag{43}$$

```
> SR[milieu][[2,3]]:=RiemannSum(-sqrt(x-2),x=2..3,method=midpoint,
  partition = n,output=sum);
  ``=expand(value(SR[milieu][[2,3]]));
  Int(-sqrt(x-2),x=2..3)=Limit('SR[milieu]'[[2,3]],n = infinity);
  S[3]:=Limit(expand(value(SR[milieu][[2,3]])),n = infinity);
```

$$\left(SR_{milieu}\right)_{[2,\,3]} := \frac{\sum_{i=0}^{n-1} \left(-\sqrt{\frac{i+\frac{1}{2}}{n}}\right)}{n}$$

$$= -\frac{\sqrt{2}\left(\sum_{i=0}^{n-1} \sqrt{\frac{2\,i}{n} + \frac{1}{n}}\right)}{2\,n}$$

$$\int_2^3 -\sqrt{-2+x} \, dx = \lim_{n \to \infty} \left(SR_{milieu}\right)_{[2,\,3]}$$

$$S_3 := \lim_{n \to \infty} -\frac{\sqrt{2}\left(\sum_{i=0}^{n-1} \sqrt{\frac{2\,i}{n} + \frac{1}{n}}\right)}{2\,n} \tag{44}$$

La mise en forme du résultat précédent nous montre que le mécanisme de la simplification automatique n'a pu obtenir la simplification de la somme formelle $SR_{milieu_{[2,\,3]}}$. En effet, nous remarquons que le symbole de sommation est bleu au lieu d'être gris. Cela nous informe que la mise en forme du résultat a été faite par le simplificateur et non pas par l'utilisateur.

```
> Sum(S[k],k=1..3)=sum(S[k],k=1..3);
```

$$\sum_{k=1}^{3} S_k = \lim_{n \to \infty} \left(\frac{1}{8\,n^2} + \frac{3}{4}\right) + \lim_{n \to \infty} \left(-\frac{2}{3\,n^2} + \frac{2}{3}\right) + \left(\lim_{n \to \infty} -\frac{\sqrt{2}\left(\sum_{i=0}^{n-1} \sqrt{\frac{2\,i}{n} + \frac{1}{n}}\right)}{2\,n}\right) \tag{45}$$

Bien que le simplificateur n'ait pu simplifié l'expression $SR_{milieu_{[2,\,3]}}$, voyons voir si l'évaluateur pourra tout de même en évaluer la limite.

```
> op(3,rhs((45)));
```

$$\lim_{n \to \infty} -\frac{\sqrt{2}\left(\sum_{i=0}^{n-1} \sqrt{\frac{2i}{n} + \frac{1}{n}}\right)}{2n} \qquad \textbf{(46)}$$

```
> ``=value((46));
```

$$= -\frac{2}{3} \qquad \textbf{(47)}$$

Maple est donc en mesure d'évaluer la troisième limite sous cette forme non réduite.

```
> Sum(S[k],k=1..3)=sum(S[k],k=1..3);
```

$$\sum_{k=1}^{3} S_k = \lim_{n \to \infty}\left(\frac{1}{8n^2} + \frac{3}{4}\right) + \lim_{n \to \infty}\left(-\frac{2}{3n^2} + \frac{2}{3}\right) + \left(\lim_{n \to \infty} -\frac{\sqrt{2}\left(\sum_{i=0}^{n-1} \sqrt{\frac{2i}{n} + \frac{1}{n}}\right)}{2n}\right) \qquad \textbf{(48)}$$

```
> ``=value(rhs((48)));
```

$$= \frac{3}{4} \qquad \textbf{(49)}$$

Reste à formuler la réponse: $\int_{-1}^{3} f(x)\, dx = \frac{3}{4}$.

**Remarque**. $\int_{-1}^{3} f(x)\, dx$ ne donne pas l'aire de la surface comprise entre la courbe et l'axe des $x$. Ni d'ailleurs la somme $\int_{-1}^{0} (x^3+1)\, dx + \int_{0}^{2} (x^2-1)\, dx + \int_{2}^{3} \left(-\sqrt{x-2}\right) dx$. Si nous avions voulu calculer l'aire de cette surface nous aurions plutôt posé le calcul suivant

$$Aire = \left(\int_{-1}^{0} (x^3+1)\, dx - \left(\int_{0}^{1} (x^2-1)\, dx\right) + \int_{1}^{2} (x^2-1)\, dx - \left(\int_{2}^{3} \left(-\sqrt{x-2}\right) dx\right)\right) u^2$$

## Exemple 5

Obtenons la valeur de $\int_{-3}^{1} \sqrt[3]{-(3+x)\,(x-1)^2}\, dx$ en appliquant la définition de l'intégrale définie.

Développons cette intégrale définie avec une somme intégrale milieu.

```
> f:=x->surd((-(3+x)*(x-1)^2),3);
  plot([x,f(x),x=-4..2],color=navy);
```

$$f := x \mapsto \sqrt[3]{-(3+x)\cdot(x-1)^2}$$



Représentons graphiquement $\int_{-3}^{1} \sqrt[3]{-(3+x)\,(x-1)^2}\,\mathrm{d}x$.

```
> coloriage([x,f(x),x=-4..2],x=-3..1);
```



Obtenons l'expression formelle de la somme intégrale $SR_{milieu_{[-3,\,1]}}$.

```
> SR[milieu][[-3,1]]:=RiemannSum(f(x),x=-3..1,method=midpoint,
  partition = n,output=sum);
  ``=expand(value(SR[milieu][[-3,1]]));
```

$$\left(SR_{milieu}\right)_{[-3,\,1]} := \frac{4\left(\displaystyle\sum_{i=0}^{n-1} 2^{2/3}\sqrt[3]{-\frac{\left(i+\frac{1}{2}\right)\left(-4+\frac{4\left(i+\frac{1}{2}\right)}{n}\right)^2}{n}}\right)}{n} \tag{50}$$

$$= \frac{8 \left( \sum_{i=0}^{n-1} \sqrt[3]{-\frac{(2i+1)(-2n+2i+1)^2}{n^3}} \right)}{n} \quad \textbf{(50)}$$

Même si Maple n'a pu donné une expression analytiquement simplifiée de cette somme intégrale, il peut être possible que l'évaluateur puisse en calculer la limite à l'infini.

```
> Int([-(3+x)*(1-x)^2]^(1/3),x = -3 .. 1)=Limit(SR[milieu][[-3,1]],
  n=infinity);
  ``=limit(rhs((50)),n = infinity);
```

$$\int_{-3}^{1} \left[-(3+x)(1-x)^2\right]^{1/3} dx =$$

$$\lim_{n \to \infty} \frac{4 \left( \sum_{i=0}^{n-1} 2^{2/3} \sqrt[3]{-\frac{\left(i+\frac{1}{2}\right)\left(-4+\frac{4\left(i+\frac{1}{2}\right)}{n}\right)^2}{n}} \right)}{n}$$

$$= \lim_{n \to \infty} \frac{8 \left( \sum_{i=0}^{n-1} \sqrt[3]{-\frac{(2i+1)(-2n+2i+1)^2}{n^3}} \right)}{n} \quad \textbf{(51)}$$

Cette fois, Maple a échoué dans le calcul de cette limite. On ne peut donc calculer $\int_{-3}^{1} \sqrt[3]{-(3+x)(x-1)^2}\, dx$ en appliquant la définition de l'intégrale définie avec une somme intégrale milieu. Par contre, nous pouvons obtenir des estimations numériques de cette intégrale définie avec des sommes intégrales infsup. Utilisons donc des sommes intégrales inférieures et supérieures de Riemann avec une partition régulière de plus en plus fine qui donneront des intervalles d'approximations de plus en plus serrés autour de la valeur de $\int_{-3}^{1} \sqrt[3]{-(3+x)(x-1)^2}\, dx$.

```
> Digits:=20;
```

$$Digits := 20 \quad \textbf{(52)}$$

```
> RiemannSum(f(x), x = -3 .. 1, method = lower, output = value,
  partition = 6);
  RiemannSum(f(x), x = -3 .. 1, method = upper, output = value,
  partition = 6);
```

$$-7.3348622189$$
$$-4.5128159043 \quad \textbf{(53)}$$

```
> plotriemann([x,f(x),x=-4..2],x=-3..1,n=6,infsup,calcul=oui);
```

$$SR_{inf_{[-3,\,1]}} = -7.3348622189$$

$$SR_{sup_{[-3,\,1]}} = -4.5128159043$$




On peut affirmer que $-7{,}3348622189 \; < \; \int_{-3}^{1} \sqrt[3]{-(3+x)\,(x-1)^2}\, \mathrm{d}x \; < \; -4{,}5128159043$. De sorte que n'importe quelle valeur de l'intervalle $[-7,3348622189;\; -4,5128159043]$ constituera une approximation de cette intégrale définie avec une erreur maximale E = $|-4.5128159043 + 7.3348622189|$ = $2.8220463146$. On peut évidemment diminuer cette erreur en raffinant davantage la partition de l'intervalle $[-3, 1]$.

```
> RiemannSum(f(x), x = -3 .. 1, method = lower, output = value,
  partition = 12);
  RiemannSum(f(x), x = -3 .. 1, method = upper, output = value,
  partition = 12);
```

$$-6.9549685728$$

$$-5.5439454155 \qquad \textbf{(54)}$$

```
> plotriemann([x,f(x),x=-4..2],x=-3..1,n=12,infsup,calcul=oui);
```

$$SR_{inf_{[-3,\,1]}} = -6.9549685728$$

$$SR_{sup_{[-3,\,1]}} = -5.5439454155$$



```
> RiemannSum(f(x), x = -3 .. 1, method = lower, output = value,
  partition = 24);
  RiemannSum(f(x), x = -3 .. 1, method = upper, output = value,
  partition = 24);
```

$$-6.7255581369$$
$$-6.0200465583 \quad \textbf{(55)}$$

```
> plotriemann([x,f(x),x=-4..2],x=-3..1,n=24,infsup,calcul=oui);
```

$$SR_{inf_{[-3,\,1]}} = -6.7255581369$$

$$SR_{sup_{[-3,\,1]}} = -6.0200465583$$




```
> RiemannSum(f(x), x = -3 .. 1, method = lower, output = value,
  partition = 48);
  RiemannSum(f(x), x = -3 .. 1, method = upper, output = value,
  partition = 48);
```

$$-6.5961410252$$
$$-6.2433852359 \quad \textbf{(56)}$$

```
> plotriemann([x,f(x),x=-4..2],x=-3..1,n=48,infsup,calcul=oui);
```

$$SR_{inf_{[-3,\,1]}} = -6.5961410252$$

$$SR_{sup_{[-3,\,1]}} = -6.2433852359$$



Avec une partition régulière en 48 sous-intervalles, on obtient l'estimation

$$-6.5961410252 \;<\; \int_{-3}^{1} \sqrt[3]{-(3+x)\,(x-1)^2}\,\mathrm{d}x \;<\; -6.2433852359$$

De sorte que n'importe quelle valeur de l'intervalle $[-6.5961410252,\ -6.2433852359]$ constitura une estimation de cette intégrale définie avec une erreur maximale E = $|-6.2433852359 + 6.5961410252|$ = $0.3527557893$.

En utilisant des sommes intégrales infsup, nous bornons donc la valeur de $\int_{-3}^{1} \sqrt[3]{-(3+x)\,(x-1)^2}\,\mathrm{d}x$.

En raffinant de plus en plus la partition de l'intervalle d'intégration, on diminuera alors l'erreur maximale commise et donc, nous obtiendrons de plus en plus un nombre de chiffres exacts dans les estimations successives. D'une part, nous ne pourrons pas déterminer à l'avance la finesse de la partition requise afin d'obtenir un nombre désiré de chiffres exacts pour l'approximation et, d'autre part, les calculs nécessaires pour obtenir un nombre satisfaisant de chiffres exacts sont assez énormes.

Mentionnons l'existence d'autres méthodes d'approximations numériques pour des fonctions continues que sont la méthode du trapèze, la méthode de Simpson et la méthode de Romberg. Ces méthodes ne seront pas exposées dans ce document. Afin d'obtenir un nombre satisfaisant de chiffres exacts, disons tout de même, que le calcul requis par l'une ou l'autre de ces méthodes est, de loin, beaucoup moins important. Ajoutons aussi, qu'avec une de ces méthodes, il est possible, *sous certaines conditions*, de borner analytiquement l'erreur commise. Ce qui permet alors, d'être plus rapidement confiant quant au nombre de chiffres exacts pour l'approximation.

La dernière section de cette feuille présente une méthode complètement différente qui est utilisée pour estimer également la valeur d'une intégrale définie. C'est la méthode de Monte Carlo. Cette méthode applique une approche probabiliste de l'estimation et est basée sur l'interprétation de l'intégrale définie comme calcul d'aire.

## Exemple 6

Dans cette exemple, nous allons comparer la valeur de $\int_1^4 \left(25-(x+2)^2\right)\,dx$ obtenue analytiquement et les approximations successives obtenues en utilisant des partitions aléatoires.

Commençons par obtenir, en premier, $\int_1^4 \left(25-(x+2)^2\right)\,dx$ avec un calcul formelle de la somme intégrale $SR_{milieu_{[1,4]}}$ avec une partition régulière de l'intervalle [1, 4]. Ensuite, nous allons créer un tableau de valeurs de sommes intégrales $SR_{aléatoire_{[1,4]}}$ calculées avec des partitions générées aléatoirement de plus en plus fines (…probablement non régulières) de l'intervalle [1, 4] et où, pour chaque partition aléatoire, le représentant $c_i$ dans chaque sous-intervalle $\left[c_{i-1}, c_i\right]$ sera choisi aussi aléatoirement.

Créons la fonction $f$.

```
> f:=x->25-(x+2)^2;
```

$$f := x \mapsto 25-(x+2)^2 \tag{57}$$

Représentons graphiquement $\int_1^4 \left(25-(x+2)^2\right)\,dx$.

```
> coloriage([x,f(x),x=-1..5],x=1..4);
```




Obtenons formellement la valeur $\int_1^4 \left(25-(x+2)^2\right)\,dx$.

```
> n:='n':
  SR[milieu][[1,4]]:=RiemannSum(f(x), x = -1 .. 4, method =
  midpoint, output = sum, partition = n);
  ``=expand(value(SR[milieu][[1,4]]));
```

$$\left(SR_{milieu}\right)_{[1,4]} := \frac{5\left(\sum_{i=0}^{n-1}\left(25-\left(1+\frac{5\left(i+\frac{1}{2}\right)}{n}\right)^2\right)\right)}{n} \tag{58}$$

$$= \frac{160}{3} + \frac{125}{12\,n^2} \quad \textbf{(58)}$$

Reste à évaluer la limite.

```
> Int(f(x),x=1..4)=Limit('SR[milieu]'[[1,4]],n = infinity);
  ``=Limit(expand(value(SR[milieu][[1,4]])),n = infinity);
  value(%);
```

$$\int_1^4 \left(25 - (x+2)^2\right) \mathrm{d}x = \lim_{n \to \infty} \left(SR_{milieu}\right)_{[1,\,4]}$$

$$= \lim_{n \to \infty} \left(\frac{160}{3} + \frac{125}{12\,n^2}\right)$$

$$= \frac{160}{3} \quad \textbf{(59)}$$

Proposons trois exemples du calcul de $SR_{aléatoire_{[1,\,4]}}$ avec des partitions générées aléatoirement et dont le représentant $c_i$ est également choisi aléatoirement dans chaque sous-intervalle créé par la partition aléatoire.

```
> a:=1;
  b:=4;
```

$$a := 1$$

$$b := 4 \quad \textbf{(60)}$$

```
> n:=4;
  P_aléa:=sort([a,stats[random,uniform[a,b]](n-1),b]);
```

$$n := 4$$

$$P\_aléa := [1, 1.3498453691, 1.7168704389, 2.3370565174, 4] \quad \textbf{(61)}$$

```
> plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,aléatoire,calcul=oui);
```

$$SR_{aléatoire_{[1,\,4]}} = \sum_{i=1}^{4} \left(25 - \left(c_i + 2\right)^2\right) \Delta x_i$$

$$= 1.0425845470$$



```
> n:=12:
  P_aléa:=sort([a,stats[random,uniform[a,b]](n-1),b]);
```

$P\_aléa := [1, 1.1840812664, 1.6927806397, 1.7232175106, 2.1210070531, 2.3895994267, 2.6277212799, 3.0242013239, 3.0259029730, 3.2070801089, 3.8153765023, 3.8992644860, 4]$ **(62)**

```
> plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,aléatoire,calcul=oui);
```

$$SR_{aléatoire_{[1,\,4]}} = \sum_{i=1}^{12} \left(25 - \left(c_i + 2\right)^2\right) \Delta x_i$$
$$= 12.5634277365$$



```
> n:=24:
  P_aléa:=sort([a,stats[random,uniform[a,b]](n-1),b]);
  plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,aléatoire,calcul=oui);
  n:='n':
```

$P\_aléa := [1, 1.0472504052, 1.1127030025, 1.1765840205, 1.2382212444, 1.3358517253,$

1.5876921802, 1.6085782508, 1.6378330824, 1.7902697550, 2.0549101768, 2.0815078344, 2.1357038209, 2.4250193557, 2.4599902429, 2.5657026626, 2.6275980278, 2.7130795579, 2.9235830009, 2.9626608925, 3.3093918937, 3.5801374266, 3.8530035598, 3.8931078511, 4]

$$SR_{aléatoire_{[1,\,4]}} = \sum_{i=1}^{24} \left(25 - \left(c_i + 2\right)^2\right) \Delta x_i$$

$$= 11.7850925161$$




Maintenant, obtenons un tableau de sommes intégrales $SR_{al\square atoire_{[1,\,4]}}$ avec une partition aléatoire différente pour chaque raffinement de la partition de l'intervalle [1,4].

```
> printf(`\n          Intégration avec une somme intégrale`);
  printf(`\n                 (Partition aléatoire)`);
  printf(`\n       |        n     |   SR[aléatoire][[1,4]]   |\n`);
  printf(`       ===========================================\n`);
   for k in [4,8,16,32,64,128,256,512,1024,2048]
   do
   P_aléa:=sort([a,stats[random,uniform[a,b]](k-1),b]):
   printf(`       |    %5d     |     %15.10f      |\n`,
   k,
   plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,aléatoire,calcul=seul)
  )
  od:
  k:='k':
```

```
          Intégration avec une somme intégrale
                 (Partition aléatoire)
       |        n     |   SR[aléatoire][[1,4]]   |
       ===========================================
       |        4     |       12.9637707433      |
       |        8     |       17.5256018117      |
       |       16     |       12.8866852785      |
```

```
|      32     |      12.6149938484      |
|      64     |      12.2004472285      |
|     128     |      12.0955707702      |
|     256     |      11.9734117923      |
|     512     |      12.0075961316      |
|    1024     |      12.0001897137      |
|    2048     |      12.0001972314      |
```

Vérifions, juste pour voir, si la valeur obtenue avec la partition aléatoire en 2048 sous-intervalles est comprise entre la somme inférieure et supérieure de Riemann.

```
> plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,infsup,calcul=seul);
```

$$SR_{inf_{[1,\,4]}} = 11.9605129186$$

$$SR_{sup_{[1,\,4]}} = 12.0394741823 \tag{63}$$

Obtenons un second tableau où, cette fois, le représentant choisi sera systématiquement le milieu de chaque partition aléatoire. Ce qui permettra d'accélérer les calculs.

```
> printf(`\n          Intégration avec une somme intégrale`);
  printf(`\n                 (Partition aléatoire)`);
  printf(`\n       |        n     |     SR[milieu][[1,4]]    |\n`);
  printf(`      ===========================================\n`);
   for k in [4,8,16,32,64,128,256,512,1024,2048]
   do
   P_aléa:=sort([a,stats[random,uniform[a,b]](k-1),b]):
   printf(`       |    %5d     |     %15.10f      |\n`,
   k,
   plotriemann([x,f(x),x=-2..5],x=1..4,P_aléa,milieu,calcul=seul))
  od:
  k:='k':
```

```
          Intégration avec une somme intégrale
                 (Partition aléatoire)
|        n     |     SR[milieu][[1,4]]    |
===========================================
|       4     |      12.3909423319      |
|       8     |      12.2215867856      |
|      16     |      12.0247362189      |
|      32     |      12.0058719095      |
|      64     |      12.0019682825      |
|     128     |      12.0008572793      |
|     256     |      12.0001840007      |
|     512     |      12.0000570539      |
|    1024     |      12.0000132197      |
|    2048     |      12.0000034646      |
```

Il semble donc qu'en choisissant systématiquement le milieu de chaque sous-intervalle des partitions aléatoires, les approximations semblent "meilleures".

Dans le cas du calcul d'une intégrale définie où une solution analytique peut être obtenue, le recours à des calculs numériques des sommes intégrales infsup ou milieu ne brille pas pour leur efficacité. Mais, par contre, dans le cas où aucun développement analytique n'est trouvée (soit formellement avec la définition ou, comme nous le verrons plus tard, soit avec le théorème fondamental du calcul), revenir à la définition de l'intégrale définie avec des calculs numériques de sommes intégrales peut être pratique pour obtenir des approximations. (Il sera, dans ce cas, préférable de considérer des sommes intégrales milieu avec des partitions régulières).

Ainsi, il est « probablement » préférable de générer des approximations de $\int_1^4 \left(25 - (x+2)^2\right) \mathrm{d}x$ avec des partitions régulières. Probablement préférable dans le sens que l'erreur commise

$$E = \left| \int_1^4 \left(25 - (x+2)^2\right) \mathrm{d}x - SR_{milieu_{[1,\,4]}} \right|$$

serait probablement plus petite.

```
> printf(`\n         Intégration avec une somme intégrale`);
  printf(`\n               (Partition régulière)`);
  printf(`\n      |       n     |     SR[milieu][[1,4]]    |\n`);
  printf(`      ==========================================\n`);
   for k in [4,8,16,32,64,128,256,512,1024,2048]
   do
   printf(`      |    %5d     |     %15.10f      |\n`,
   k,
   plotriemann([x,f(x),x=-2..5],x=1..4,n=k,milieu,calcul=seul))
  od:
  k:='k':
```

```
         Intégration avec une somme intégrale
               (Partition régulière)
      |       n     |     SR[milieu][[1,4]]    |
      ==========================================
      |       4     |       12.1406250000      |
      |       8     |       12.0351562500      |
      |      16     |       12.0087890625      |
      |      32     |       12.0021972656      |
      |      64     |       12.0005493164      |
      |     128     |       12.0001373291      |
      |     256     |       12.0000343323      |
      |     512     |       12.0000085831      |
      |    1024     |       12.0000021458      |
      |    2048     |       12.0000005364      |
```

Ces résultats semblen montrer que les estimations convergent vers 12.

# Intégration numérique et expérience de Monte Carlo

L'exemple 6 de la section précédente a su montrer comment l'ordinateur pourrait être utile pour évaluer approximativement une intégrale définie lorsqu'il ne serait pas possible d'obtenir un développement analytique. Malgré que nous n'ayons pas présenté les méthodes d'approximations que sont les méthodes du trapèze, de Simpson et de Romberg, disons quand même que leur efficacité repose aussi sur l'utilisation de l'ordinateur. Mais, dans le cas de fonctions présentant de grandes oscillations, l'utilisation d'une de ces méthodes (trapèze, Simpson, Romberg) s'avère difficile d'application. La méthode de Monte Carlo constitue alors une alternative intéressante d'obtenir, dans de tels cas, des approximations d'une intégrale définie.

Soit la fonction *f* définie par $f(x) = x(1-x)\left(\sin^2\right)(100x(1-x))$ définie sur l'intervalle [0,1].

```
> f:=x->x*(1-x)*(sin^2)(100*x*(1-x));
```

$$f := x \mapsto x\cdot(1-x)\cdot\left(\sin^2\right)(100\cdot x\cdot(1-x)) \quad \textbf{(64)}$$

```
> Surface:=coloriage([x,f(x),x=0..1],x=0..1,numpoints=120,view=[0..1,0..0.25]):
  Surface;
```




Obtenons $\int_0^1 x(1-x)\left(\sin^2\right)(100x(1-x))\,dx$ en appliquant la méthode de Monte Carlo.

En interprétant l'intégrale définie à calculer comme un calcul d'aire, imaginons que la surface entre la courbe et l'axe des *x* est une portion de la surface du rectangle représenté ci-dessous.

```
> Rectangle:=plottools[rectangle]([0,0.25], [1,0],color = white,
  linestyle=4):
  plots[display]([Surface,Rectangle],axes=none);
```



La méthode de Monte Carlo est, somme toute, assez simple. À l'aide du générateur aléatoire *random* (pseudo aléatoire), nous allons générer des points $[x_i, y_i]$, $i = 1, 2, 3, ...,$ n à l'intérieur de ce rectangle. Ensuite, sur la base probabiliste de la dispersion des points à l'intérieur de ce rectangle, on posera que

$$\frac{\textit{Aire sous la courbe}}{\textit{Aire du rectangle}} \approx \frac{\textit{nombre de points sous la courbe}}{\textit{nombre total de points généré}}$$

d'où

$$\textit{Aire sous la courbe} = \int_0^1 x\,(1-x)\,\left(\sin^2\right)(100\,x\,(1-x))\,\mathrm{d}x \approx \textit{Aire du rectangle} \times \frac{\textit{nombre de points sous la courbe}}{\textit{nombre total de points}}$$

Créons la procédure Monte_Carlo qui générera aléatoirement un nombre *n* de points dans le pavé [a, b] X [c,d]. En plus d'une liste de *n* points qui seront générés, il sera possible de visualiser ces points en les superposant aux tracé du rectangle et de la région colorée sous la courbe.

```
> Monte_Carlo:=proc(n::posint,Tracé::name)
   local Nuage,X,Y;
   global Points;
   X:=[stats[random,uniform[a,b]](n)]:
   Y:=[stats[random,uniform[c,d]](n)]:
   Points:=zip((x,y)->[x,y],X,Y):
   if Tracé=oui then
    Nuage:=plot(Points,style=point,symbol=circle,color=black):;
    plots[display]([Surface,Rectangle,Nuage])
    else Points
   fi
  end:
```

Posons les bornes *a* et *b* qui délimiteront les valeurs d'abscisses qui seront générées: $a \le x_i \le b$.
De même pour les bornes *c* et *d* qui délimiteront les valeurs d'ordonnées qui seront générées: $c \le y_i \le d$.

```
> a:=0:
  b:=1:
  c:=0:
  d:=0.25:
```

Utilisons la macro-commande Monte_Carlo avec $n = 600$ et précisons le second argument "*oui*" pour obtenir la représentation graphique du nuage de points superposé aux tracé du rectangle et de la région colorée sous la courbe.

```
> Monte_Carlo(600,oui);
```




Il reste ensuite à inclure le calcul de

$$\textit{Aire du rectangle} \times \frac{\textit{nombre de points sous la courbe}}{\textit{nombre total de points}}$$

dans une petite procédure pour être en mesure d'obtenir l'approximation cherchée.

```
> approximation:=proc()
   local k,P,Compteur;
    Compteur:=0;
    for k from 1 to nops(Points) do
     if Points[k,1]>=0 and Points[k,2]<f(Points[k,1]) then
      Compteur:=Compteur+1
     fi
    od;
   evalf(abs(b-a)*abs(d-c)*Compteur/nops(Points))
  end:
```

Avec la macro-commande approximation, obtenons une première approximation de $\int_0^1 x(1-x)\left(\sin^2\right)(100x(1-x))\,dx$ avec les 300 points précédemment générés.

```
> approximation();
```

$$0.0720833333 \quad \textbf{(65)}$$

Présentons dans un tableau les approximations successives de $\int_0^1 x(1-x)\left(\sin^2\right)(100x(1-x))\,dx$ obtenues avec un nombre $n$ de points de plus en plus grand.

```
> printf(`\n       Intégration avec Monte Carlo`);
  printf(`\n       |    n   |   Approximation  |\n`);
  printf(`      =============================\n`);
   for k in [100,200,300,400,500,1000,2000,3000,4000,5000,10000,20000,
  30000]
```

```
  do
  Monte_Carlo(k,non);
  printf(`       | %5d  |   %10.10f   |\n`,
  k,
  approximation())
 od:
 k:='k':
```

```
     Intégration avec Monte Carlo
     |    n   |   Approximation  |
     ============================
     |   100  |   0.0725000000   |
     |   200  |   0.0662500000   |
     |   300  |   0.0691666667   |
     |   400  |   0.0837500000   |
     |   500  |   0.0725000000   |
     |  1000  |   0.0812500000   |
     |  2000  |   0.0752500000   |
     |  3000  |   0.0725833333   |
     |  4000  |   0.0742500000   |
     |  5000  |   0.0714500000   |
     | 10000  |   0.0762750000   |
     | 20000  |   0.0747125000   |
     | 30000  |   0.0750583333   |
```

En se basant sur les résultats précédents, il est raisonnable de « croire » que $\int_0^1 x(1-x)\left(\sin^2\right)(100x(1-x))\,\mathrm{d}x \approx 0{,}075$. Mais, qu'en est-t-il de la précision de cette approximation ? Ces trois chiffres de la partie fractionnaire sont-ils exacts ? La méthode de Monte Carlo ne permet pas, non plus, de borner l'erreur commise. Alors, il faut certainement privilégier la recherche d'une solution analytique au calcul d'une intégrale définie. À défaut d'une solution analytique, nous pouvons toujours avoir recours à cette méthode qui peut même être appliquer à des fonctions de plusieurs variables.

Notons que Maple est en mesure de formuler une solution analytique de $\int_0^1 x(1-x)\left(\sin^2\right)(100x(1-x))\,\mathrm{d}x$, mais, cela passe par des fonctions intégrales de Fresnel, lesquelles fonctions ne sont pas des modèles dont nous prendrons connaissance dans notre cours... mais, tout de même, il existe une solution analytique pour cette intégrale définie.

```
> Int(f(x),x=0..1)=int(f(x),x=0..1);
  ``=evalf(rhs(%));
```

$$\int_0^1 x(1-x)\sin(100x(1-x))^2\,\mathrm{d}x = \frac{1}{12} - \frac{\sqrt{\pi}\cos(50)\operatorname{FresnelC}\left(\frac{10}{\sqrt{\pi}}\right)}{80} - \frac{\sqrt{\pi}\cos(50)\operatorname{FresnelS}\left(\frac{10}{\sqrt{\pi}}\right)}{8000} + \frac{\sqrt{\pi}\operatorname{FresnelC}\left(\frac{10}{\sqrt{\pi}}\right)\sin(50)}{8000}$$

$$-\frac{\sqrt{\pi}\,\sin(50)\,\mathrm{FresnelS}\left(\frac{10}{\sqrt{\pi}}\right)}{80}$$

$$= 0.0754391579 \qquad \textbf{(66)}$$